
BY MARK HEDGES 

PART I explored several new
methods of setting up

program housekeeping using ESA/370
and z/OS instructions. This concluding
article will explore the use of several
facilities in IBM’s High Level Assembler
(HLA) that make assembler code easier
to design and maintain.

ASSEMBLER EVOLUTION

IBM’s HLA, now at release 4, provides
facilities to improve code readability and
to identify assembly errors faster. The
new assembler expands on what
Assembler-H provided, long variable
and label names. With HLA, long vari-
able names and lower-case support was
added to the macro processor as well.
Who ever liked trying to choose a good
name for a label when only eight-
character names were allowed? HLA
also provides for a structure called
“labeled USINGs.” With a labeled
USING you can have two registers map
the same DSECT name and avoid the
sometimes confusing indirect addressing
that you would normally have to imple-
ment when you wanted to map the same
storage area. Last, but certainly not
least, HLA also has vastly improved
diagnostics that, for example, determine
addressing limitations.

This article will explain the use of
several of these new HLA facilities and

explore the new HLA/ToolKit struc-
tured macros through several examples.

LABELED USING EXAMPLE

A labeled USING allows your pro-
grams to directly relate a named
storage variable to a specific USING
statement. When you establish a
USING, place a label in column one.
Then, to refer to any storage variable
name mapped by the USING state-
ment, add a prefix to the reference
using the label name followed by a
period. The example in Figure 1
demonstrates the mapping of the CVT
through labeled USINGs.

Figure 2 provides an example of a
labeled USING in a macro that obtains
system information from the extended
CVT. An assembler macro variable
‘&U1’ is defined using the ‘&SYSNDX’
built-in counter for use in the labeled
USING. &SYSNDX always generates
a unique number so this notation pro-
vides for multiple invocations of the
macro. I suggest using labeled USING
in this fashion in all macros. This
implementation will resolve a common
problem with non-labeled USINGs par-
ticularly in macros. The question and
occasional problem I ran into is “What
happens when a DROP statement is
issued when multiple USINGs are used

L R15,CVTPTR
@CVT    USING CVTMAP,R15

TM @CVT.CVTOSLV3,CVTZOS_V1R2 Are we z/OS 1.2 or higher?
BO ZOSV12                     Yes, go do something

DROP @CVT     Drop the labeled USING reference

FIGURE 1: LABELED USING 

Macro
MVSREL  &Product=,&Version=,&release=,&modification=

&U1 SETC   ‘U1&SYSNDX.’
L     R1,CVTPTR                    Get CVT
L     R1,CVTECVT-CVTMAP(,r1)        Extended CVT

&U1  USING ECVT,R1                      Map the ECVT
MVC   &Product.(L’ECVTPNAM),&U1..ECVTPNAM    Product name
MVC   &Version.(L’ECVTPVER),&U1..ECVTPVER    Product version
MVC   &Release.(L’ECVTPREL),&U1..ECVTPREL     Product Release
MVC   &Modification.(L’ECVTPMOD),&U1..ECVTPMOD Product mod level
DROP  &U1

.MEND MEND

FIGURE 2: LABELED USING IN A MACRO 

for the same DSECT?” Labeled USINGs
solve that dilemma by making a direct
connection when a USING is established
and dropped.

THE HLA TOOLKIT

At about the same time as the HLA was
released, IBM offered a charge-able featured
called the High Level Assembler ToolKit
Feature. This product has some nice debug-
ging facilities and a great set of supported
structured assembler macros. Several
macros in the ToolKit, such as DO, SELECT
and IF/THEN/ELSE structures make writing
and maintaining code easier.

HLA TOOLKIT SELECT
MACRO EXAMPLE

How many different methods can you
use to perform bit testing when evaluating
flag settings? The answer is many.
Typically, you can use the assembler Test
Under Mask (TM) instruction to evaluate
bit settings. The TM instruction sets the
condition code that you can test using a
subsequent branch mnemonic like Branch
on Ones (“BO”). When you need to test a
flag for multiple bit settings, the visual path
length of the code can seem long and con-
fusing. Code readability problems can arise
due to a programmer’s inability to choose
how he implements the combination of the
TM and associated condition code checking
and the number of bit tests. Can this situation
be avoided?

Figure 3 shows a “traditional” way of
coding an exclusive bit test for a value
called FLAG. If a match is found, the “BO”
instructs the program to branch to the
appropriate routine. If the value in FLAG is
set to X’01’ (BitOne), then processing is
done for that condition. The problem here is
a visual one: There are too many lines of
code to pour through to determine the exe-
cution logic.

The second example in Figure 3 presents
another way of coding the same exclusive
test with less lines of code. Direct branching
is used (BNO mnemonic) if a match is not
found to skip to perform the next bit test.
This method works but does not follow
good coding practices.

The third example in Figure 3 uses the
IBM SELECT structured macro. The gener-
ated code results in the same execution as
the other two samples in Figure 3 and its

readability is much improved. The SELECT
macro is very powerful when you are
required to perform an exclusive test for bit
and character strings.

You can extend the use of the SELECT
function in many ways to assist in testing

multiple conditions, such as determining
the release of MVS. Figure 4 presents an
example using a nested SELECT macro.
You can use these nested structures as
many times as necessary as long as you
provide a matching ENDSEL. The first

� The value “Flag” is set to a unique value.
Flag DS    XL01’00’
BitOne equ      X’01’
BitTwo equ      X’02’
* Test exclusive settings of FLAG 

TM  Flag,Bitone
BO BitOne_Set          Bit One set
TM  Flag,BitTwo
BO  BitTwo_Set         Bit Two set
Bas R14,NoBitsSet       No bit was set that we care about
B  Skip_Bit_Processing   Skip code

Bitone_Set EQU *
BAS R14,BitOne_Routine
B Skip_Bit_Processing

Bittwo_Set EQU *
BAS R14,BitTwo_Routine
B Skip_Bit_Processing

Skip_Bit_Processing  EQU  *
� Another example of performing the same operation as above.
*  Though the following style is considered “Bad” programming it
*  accomplishes the same result as the lines of code above
*  I like this because it is rather easy to follow and allows 
*  for the addition of new bit set testing by simply replicating
*  a few lines.

TM  Flag,Bitone
BNO *+12               Skip if bit not set
BAS R14,BitOne_Routine
B   Skip_bit_Processing
TM  Flag,BitTwo
BNO *+12             Skip if bit not set
BAS R14,BitTwo_Routine
B   Skip_bit_Processing
Bas R14,NoBitsSet

* All done here
B  Skip_Bit_Processing

Skip_Bit_Processing   EQU  *

� The easiest and most straight forward method using the HLA/Tool-Kit.
COPY  ASMMSP                HLA/ToolKit        
Select  TM,FLAG,O    Test to see if the FLAG Bit is set ON

When  (BitOne)
BAS   R14,BitOne_Routine

When  (BitTwo)
BAS   R14,BitTwo_Routine

Othrwise   ,         None of the above
Bas   R14,NoBitsSet

EndSel

� - This example uses the “TM” (Test Under Mask) instruction and “BO” (Branch if Ones) mnemonic when a match is
discovered. 

� - This example uses the “TM” (Test Under Mask) instruction and “BNO” (Branch if Not Ones) mnemonic when a
match is not present.  The “BNO *+12” causes the program to skip to the next “TM instruction if there is no match.

� - This example uses the HLA/Toolkit’s “Select” (a TM with a “BO”) and then tests for specific matching conditions.
This saves a time in identifying the programmer’s intent and the ability to maintain the code.

FIGURE 3: BIT TESTING USING THE HLA TOOLKIT

Select   TM,CVTOSLV3,O                           
When  (X’10’)               z/OS 1.2           

BAS  R14,DO_Zos12        
When  (X’20’)               z/OS 1.1           

BAS  R14,DO_Zos11        
Othrwise                                       

Select  TM,CVTOSLV2,O     OS/390 R2          
When  (X’01’)          OS/390 2.10        

BAS  R14,OS390_210 
When  (X’02’)          OS/390 2.9         

BAS  R14,OS390_29
Othrwise  ,               OS/390 r1          

Select  TM,CVTOSLV1,O                   
...etc.

EndSel
EndSel

FIGURE 4: OS LEVEL CHECKING 

section of the example may be the old way
of performing the test followed by the use
of the SELECT macro.

HLA TOOLKIT IF
MACRO EXAMPLE

The IF/ELSE/ENDIF structured macros
in the HLA ToolKit simplify complicated
condition checking when attempting to val-
idate multiple conditions. A good example
of such a situation is when you need to
check the contents of a passed parameter
list. When you code a JCL EXECute state-
ment using a PARM= parameter, the value
specified in the PARM is passed to the
called program. Then, it becomes the
responsibility of the program to process any
of the PARM parameters if desired. Figure
5 provides such an example where the routine
validates the presence of a PARM= value
and uses structured macros to validate and
process expected values.

HLA TOOLKIT DO
MACRO EXAMPLE

The DO macro has many possible options
including a sub-parameter of “INF” to exe-
cute the DO loop infinitely. The “DO INF”
is handy for processing an input file until
you reach end-of-file. The example in
Figure 6 processes an input file called
SYSIN until EOF is reached. Notice that
this program operates in AMODE 31 and
uses MODE31 and MODE24 macros to
switch back and forth between addressing
modes to accommodate the requirement for
below-the-line I/O.

The first example in Figure 7 presents a
basic form of the DO macro using the
“FROM” parameter. The DO loop is exe-
cuted until the value in R5 is zero. It
would be a good idea for you to ensure
that R5 is positive before you execute the
DO loop.

The second example in Figure 7 presents
the DO macro using the “UNTIL” clause.
The logic in this sample is more complex
than the DO UNTIL clause. Notice the
DOEXIT. The DOEXIT says to exit the DO
loop if the condition is met; in this case, that
the contents of R3 are zero. You can use
multiple DOEXIT verbs throughout a DO
loop to terminate execution. The purpose of
this sample is to process all dynamic LPA
modules. The operating system release level
is tested, and then the DO loop is performed

until all modules in the dynamic LPA are
processed or until the next module in the
chain is zero (via the DOEXIT macro).

HLA TOOLKIT CASENTRY

MACRO EXAMPLE

The CASENTRY macro, in particular,
simplifies the processing of return codes
from a called service or program. Many

Bakr  r14,0                Save environment on the linkage stack
* other house keeping

ereg r0,r1                load r0, r1 from stack
SLR r3,r3              Clear R3
L  R2,00(,R1)         =A(=L’PARM =A(PARM)))

* Load a half word into R3 which is the length of the passed PARM
If   (ICM,R3,B’0011’,0(r2),nz),and,      Any parm at all?       +

(c,r3,le,=AL04(l’Parm))            Is it too big?
SLR  R15,R15
ST R3,PARM_LEN           We have a PARM value
BCTR R3,00                for EX
MVC PARM(*-*),00(R2)      Copy the PARM to our storage
EX  R3,*-06              (really do it)
OC  PARM,Blanks         Upper case PARM

*  It would be nice to use the SELECT macro, but we are unsure of 
*  the length of the PARM value

IF  (clc,=C’HELLO’,eq,parm)
BAS R14,DO_HELLO
B  END_PARM

EndIF 
IF (clc,=C’GOODBYE’,eq,parm)

BAS R14,DO_GOODBYE
B  END_PARM

EndIF 
Else  ,

LA  r15,08
EndIf ,

END_PARM EQU   *
* The data section should contain:
Blanks  dc   CL100’ ‘

DS  0F
PARMLen DS  AL04(L’Parm)
Parm    DS   CL60                    maximum size is 100 we want just 60 bytes 

FIGURE 5: IF LOGIC TO VALIDATE THE VALUE AND PRESENCE OF A PARM=VALUE 

OPEN  (SYSIN,INPUT),MODE=31
DO   INF               Loop until EOF

MODE24  ,             change to AMODE 24
GET  SYSIN,File_area   Get a record
MODE31  ,             back to AMODE 31
BAS  R14,Process_SYSIN

ENDDO 
EOF_SYSIN  equ  *

MODE31  ,              Need to reset to MODE 31
CLOSE  (SYSIN),MODE=31

* Data Section
SYSIN   DCB DDNAME=SYSIN,MACRF=GL,DSORG=PS,LRECL=80,BLKSIZE=800,

RECFM=FB,EODAD=EOF_SYSIN,EXLST=0

FIGURE 6: DO MACRO WITH “INF” PARAMETER 

(1)     L   R5,Full_word
DO  FROM=(r5)       Loop

BAS  R14,Process
ENDDO 

(2)
@ECVT    USING ECVT,R2

L     r2,cvtptr
If    (TM,cvtoslv2-cvtmap(r2),CVTDYLPA,o)  Dynmaic LPA available?

l    r2,CVTECVT-CVTMAP(,r2)     =A(EVCT)
IF   (ICM,R3,B’1111’,@ECVT.ECVTCDYN,NZ)  Start of dynamic LPA

Do   UNTIL=(C,R3,EQ,@ECVT.ECVTCDYS)
Lr  R1,r3               Load LPA module address
bas balreg,DO_something
L  r3,00(,r3)
DOExit (ltr,r3,r3,z) Just in case MUST DO

EndDo ,
EndIf  ,

EndIf  ,
Drop  @ECVT

FIGURE 7: DO MACRO LOOP 

older programs processed return codes in
R15 using a branch table. The CASENTRY
macro obviates the need to build and
maintain a branch table. Figure 8 provides
such an example. The ability of the CASE
statement to process multiple conditions
within the same statement (like the WHEN
clause of the SELECT macro) makes it con-
venient to use.

ASSEMBLER HOUSEKEEPING

I have a final suggestion for a quick way
to standardize assembler housekeeping:
Create and use a common member that con-
tains variables and global variables you can
use in “open” code within all your programs.
Place a common member as the first file in
a SYSIN concatenated job stream, as shown
in Figure 9.

Enjoy coding in the z/OS world!  

Mark Hedges is a product manager for Allen
Systems Group (ASG) in Reston, VA. He can be
contacted via email at mhedges@landmark.com
or at (703) 464-1375.

Casentry r15,power=2                                         
Case 4                 Return code=04        

* do something
Case 8                 Return code=08               

* do something
Case 12,16,20,24                                           

EndCase                                                      

FIGURE 8: CASENTRY MACRO 

*PROCESS BATCH,XREF(SHORT,UNREFS) �
COPY ASMMSP    HLA/ToolKit
GBLC &YYYY,&MM,&DD,&ADATE  �

&YYYY SETC ‘&SYSDATC’(1,4)             Year
&MM     SETC ‘&SYSDATC’(5,2)             Month
&DD    SETC ‘&SYSDATC’(7,2)             Day
&ADATE  setc ‘&YYYY./&MM./&DD’   �

� Provide any assembler overrides. This must be the first statement in the assembly deck.
� Define all possible global variables you may want to use in any of your programs. These could include a variable that
would contain your company name
� Set the date in a format I like. I can use this modified format in the module eye-catcher.

FIGURE 9: COMMON MEMBER IN YOUR LOCAL USERMOD LIBRARY 


